
This case study looks at the re-engineering of a program to derive a non-

You can measure a programmer's perspective by noting his attitude 
on the continuing vitality of FORTRAN.

INTRODUCTION

The Towers of  Hanoi (TOH) is a classical problem in computer science, often dealing with the 
illustration of  recursion. Non-recursive solutions are less elegant, but offer another view into the 
problem space. This pracnique deals with the re-engineering of  a Fortran program from 1984, 
written to solve the TOH in a non-recursive manner. The most common non-recursive solutions 
use a stack in place of  recursion. The algorithm posed by Mayer and Perkins is dubbed the 
“Fourth Grade Solution” because it was discovered by D. Perkins whilst in the fourth grade.

THE ALGORITHM

The algorithm works in the following manner to move n disks:

1. Take an unsigned binary counter of  n bits, initialized to zero, and count up all possible 
numbers.

2. Let the rightmost bit have index 1, and the most significant have index n.
3. During each of  the 2n-1 increments, exactly one bit changes from 0 to 1. The index of  that 

bit is the index of  the disk to be moved next.
4. Function INCR (i) terminates as soon as one bit changes from 0 to 1, and (ii) returns the 

index of  that bit.

This algorithm has 2n-1 moves. 

Here is a binary counter of  n=3 bits: 23 representable values and 2n-1 increments.

0 0 0 0 0 1 0 1 0 0 1 1 1 0 0 1 0 1 1 1 0 1 1 1

Where to move the first disk? The final location for n=odd and the buffer for n=even.

How does it work? The index of  the disk to be moved is identical to the index of  the bit changing 
from 0 to 1 when counting from 0 to 2n-1.

3 2 1 index

0 0 0 bit

A binary count sequence from 0 to 2n-1 changes only one single bit from 0 to 1 during each 
increment. For example the fifth step in the binary count sequence above changes the 4 or “100” 
to 5 or “101”. The bit that flips has index 1. Here is the sequence of  flip-bits for n=3.

0 0 0 0 0 1 0 1 0 0 1 1 1 0 0 1 0 1 1 1 0 1 1 1

flip-bit
index

1 2 1 3 1 2 1

Now the flip-bits need to be stored somewhere. To determine where, Mayer and Perkins use two 
equations based on whether the value returned from the function INCR is:

1:
! to_pos := 6 - where_1_is - where_disk_is

or any other value:

! to := 6 - HOLD(1) - HOLD(DISK)

The Fortran function INCR is called 2n-1 times, each time incrementing the bit string passed. 
The function terminates when a 0 changes to 1. It returns the index of  the bit that changes from 
0 to 1. 

REF

Mayer, H., Perkins, D., “Towers of  Hanoi Revisited A Nonrecursive Surprise”, SIGPLAN Notices, 
Vol.19, pp.80-84 (1984)

THE LEGACY CODE
C      MAYER & PERKINS
C      "TOWERS OF HANOI REVISITED"

       PROGRAM HANOI
       INTEGER DISKS
10     FORMAT(' ENTER NUMBER OF DISKS WANTED ')
20     FORMAT(I10)
30     WRITE(*,10)
       READ(*,20) DISKS
       IF (DISKS .GT. 0) CALL TOWER(DISKS)
       IF (DISKS .LT. 0) GOTO 30
       END

       SUBROUTINE TOWER(TOTAL)
       INTEGER TOTAL
       INTEGER FROM, TOD, DISK, TEMP, I
       INTEGER BITSTR(10), HOLD(10)
       CHARACTER * 6 PLACE(3)
       DATA PLACE /'START ', 'BUFFER', 'FINISH'/
       DO 10 I=1, 10
       BITSTR(I) = 0
10     HOLD(I) = 1
       TEMP = 3 - MOD(TOTAL,2)
       DO 20 I=1, 2**TOTAL - 1
       DISK = INCR(BITSTR)
       IF (DISK .EQ. 1) THEN
       FROM = HOLD(1)
       TOD = 6 - FROM - TEMP
       TEMP = FROM
       ELSE
       FROM = HOLD(DISK)
       TOD = 6 - HOLD(1) - HOLD(DISK)
       END IF
       WRITE(*,*) 'MOVE DISK ',DISK,' FROM ',PLACE(FROM),
     - ' TO ',PLACE(TOD)
       HOLD(DISK) = TOD
20     CONTINUE
       END

       INTEGER FUNCTION INCR(BITSTR)
       INTEGER BITSTR(10)
       INTEGER I
       I = 1
10     IF (BITSTR(I) .EQ. 0) THEN
       BITSTR(I) = 1
       INCR = I
       RETURN
       ELSE
       BITSTR(I) = 0
       I = I + 1
       GOTO 10
       END IF
       END

SAMPLE RUN

Here is a sample run of  the program with three disks:

ENTER NUMBER OF DISKS WANTED
3
 MOVE DISK            1  FROM START  TO FINISH
 MOVE DISK            2  FROM START  TO BUFFER
 MOVE DISK            1  FROM FINISH TO BUFFER
 MOVE DISK            3  FROM START  TO FINISH
 MOVE DISK            1  FROM BUFFER TO START
 MOVE DISK            2  FROM BUFFER TO FINISH
 MOVE DISK            1  FROM START  TO FINISH

LEGACY ISSUES

The program as it stands suffers from some readability issues, partly because of  the fixed 
formatting of  older versions of  Fortran - lack of  indenting, use of  uppercase characters. This of  
course can be easily fixed. The output also suffers from some usability issues - the UPPERCASE 
text seems like the “PROGRAM IS YELLING AT ME”, and the terms START, FINISH, 
BUFFER could be replaced with more meaningful terms.

The program itself  is split up into a main part, a subroutine and a function. The main program 
contains one GOTO statement which deals with user input that is less than zero - this can easily 
be reworked. The subroutine TOWER contains two older DO loops which use labels, and an 
older predefined string. The function INCR contains one simple GOTO statement to remove. 

Overall the program will not be that challenging to re-engineer. 

FIRST THINGS FIRST

The first thing to do is change the file extension to .f95, and modify the comment delimiter from 
C to !. The file will still compile and run. (see listing hanoi.f95)

IMPROVE READABILITY

To improve readability the first task is to convert the program from uppercase to lowercase. This 
can be achieved using the “translate characters” utility, tr:

! tr '[:upper:]' '[:lower:]' < hanoi.f95 > hanoi_1.f95

Note that at the moment, the program is not format-free, this will be done after the jump labels 
have been removed, so there is some clarity during the re-engineering process. It can be indented 
as the re-engineering progresses to improve program clarity. Each major update will be provided 
in a separate file to illustrate the evolution of  the re-engineering process.

Trying to compile the program now will result in a compiler error. 

hanoi_1.f95:34.57:

       write(*,*) 'move disk ',disk,' from ',place(from),
                                                         1
Error: Expected expression in WRITE statement at (1)
hanoi_1.f95:35.6:

     - ' to ',place(tod)
      1
Error: Invalid character in name at (1)

This has occurred due to the “line continuation” operator used in the legacy program to split an 
expression over two lines (in column 6) - this is because F77 had a limit of  66 available columns 
for a statement. Any character can be used as the continuation character. To re-engineer this 
problem requires replacing the continuation operation with an &. So this code:

       write(*,*) 'move disk ',disk,' from ',place(from),
     - ' to ',place(tod)

becomes:

       write(*,*) 'move disk ',disk,' from ',place(from), &
       ' to ',place(tod)

Although as implementations of  Fortran after F77 did away with this limitation, it is just as easy 
to place this all on one line.

       write(*,*) 'move disk ',disk,' from ',place(from),' to ',place(tod)

RE-ENGINEERING VARIABLES

The first portion of  the re-engineering involves modifying variables to remove implicit 
declarations, and enhance the declaration criteria to F95 standards. Variables that are implicitly 
declared in the original program are converted to explicit declarations. For example, integer 
variables are converted in the following manner:

! integer disks

becomes

! integer :: disks

The declaration and initialization of  the character array place is modified from:

! character * 6 place(3)
! data place /'start ', 'buffer', 'finish'/

to

! character(len=6), dimension(3) :: place
! place  = (/'start ', 'buffer', 'finish'/)

RE-ENGINEERING OPERATORS

The old-fashioned operators such as .eq. should also be changed to their modern equivalent:

! .eq. becomes ==
! .gt. becomes >
! .ne. becomes /=
! .lt. becomes <

See listing hanoi_2.f95.

RE-ENGINEERING THE MAIN PROGRAM

So the question is which of  the legacy issues should be dealt with first? The best way is to focus 
on each module of  the program, starting with the main program. 

       program hanoi
       integer :: disks
10     format(' enter number of disks wanted ')
20     format(i10)
30     write(*,10)
       read(*,20) disks
       if (disks > 0) call tower(disks)
       if (disks < 0) goto 30
       end

This is relatively easy and involves:

1. Modify the I/O statements.
2. Modifying the if statements to consolidate them, and remove the goto statement and 

associated labels. 

Firstly, the if statement are consolidated. Basically there are three scenarios here:
1. if  the number of  disks entered is greater than 0, the subroutine tower is called.
2. if  the number of  disks entered is less than 0, the user is asked for the input again.
3. if  the number of  disks is zero, the program terminates (this is implied by the way the 

program is written).

The goto statement can be removed by encasing much of  the code inside a do loop. Inside the 
loop, the user is prompted for the number of  disks, then if  that value is greater than zero, control 
is passed to the subroutine tower; and if  it’s equal to zero the program exits. Any other value 
and it cycles around and asks the user for an input value again. This is what the main program 
looks like now:

       program hanoi
       integer :: disks
       do
    10     format(' enter number of disks wanted ')
           write(*,10)
           read (*,"(i10)") disks
           if (disks > 0) then
               call tower(disks)
               exit
           else if (disks == 0) then
               exit
           end if
       end do
       end

Notice that the format string for the read statement has been consolidated inside the read 
statement.

See listing hanoi_3.f95.

RE-ENGINEERING THE INCR FUNCTION

Next the function incr will be re-engineered:

       integer function incr(bitstr)
       integer :: bitstr(10)
       integer :: i
       i = 1
10     if (bitstr(i) == 0) then
       bitstr(i) = 1
       incr = i
       return
       else
       bitstr(i) = 0
       i = i + 1
       goto 10
       end if
       end

Remember here that with Fortran functions, the name of  the function also serves as the variable 
in which the value is returned. So the main legacy issue here is the goto statement. This is 
essentially a do loop disguised as a goto, so the fix is to encapsulate the if statement within a do 
loop. Easy. There is also the need to specify the type of  parameter being passed to the function, 
in this case bitstr is an inout variable (value passed in, and passed out again). Here is the 
resulting function:

       integer function incr(bitstr)
       integer, intent(inout) :: bitstr(10)
       integer :: i
       i = 1
       do
           if (bitstr(i) == 0) then
               bitstr(i) = 1
               incr = i
               return
           else
               bitstr(i) = 0
               i = i + 1
           end if
       end do
       end

See listing hanoi_4.f95.

RE-ENGINEERING THE TOWER SUBROUTINE

The final piece of  re-engineering involves the subroutine tower. This really just involves cleaning 
up the do loops - removing the continue and the labels, and adding an intention type to the 
subroutine parameter total - in this case it is a variable that is just read in to the subroutine.

       subroutine tower(total)
       integer :: total
       integer :: from, tod, disk, temp, i
       integer :: bitstr(10), hold(10)
       character(len=6), dimension(3) :: place
       place  = (/'start ', 'buffer', 'finish'/)
       do 10 i=1, 10
       bitstr(i) = 0
10     hold(i) = 1
       temp = 3 - mod(total,2)
       do 20 i=1, 2**total - 1
       disk = incr(bitstr)
       if (disk == 1) then
       from = hold(1)
       tod = 6 - from - temp
       temp = from
       else
       from = hold(disk)
       tod = 6 - hold(1) - hold(disk)
       end if
       write(*,*) 'move disk ',disk,' from ',place(from), ' to ', place(tod)
       hold(disk) = tod
20     continue
       end

Here’s what the subroutine looks like after the modifications are made:

       subroutine tower(total)
       integer, intent(in) :: total
       integer :: from, tod, disk, temp, i
       integer :: bitstr(10), hold(10)
       character(len=6), dimension(3) :: place
       place = (/'start ', 'buffer', 'finish'/)
       do i = 1,10
           bitstr(i) = 0
           hold(i) = 1
       end do
       temp = 3 - mod(total,2)
       do i = 1, 2**total - 1
           disk = incr(bitstr)
           if (disk == 1) then
               from = hold(1)
               tod = 6 - from - temp
               temp = from
           else
               from = hold(disk)
               tod = 6 - hold(1) - hold(disk)
           end if
           write(*,*) 'move disk ',disk,' from ',place(from), &
                      ' to ', place(tod)
           hold(disk) = tod
       end do
       end

See listing hanoi_5.f95.

FINAL CHANGES

The final changes have to do with improving the usability of  the program, by re-writing some of  
the text phrases. Here is the result with the appropriate changes made:

How many disks?
3
 move disk            1  from post 1 to post 3
 move disk            2  from post 1 to post 2
 move disk            1  from post 3 to post 2
 move disk            3  from post 1 to post 3
 move disk            1  from post 2 to post 1
 move disk            2  from post 2 to post 3
 move disk            1  from post 1 to post 3

There is still the issue of  the large space in the output after the work disk. This can be fixed by 
adding a format statement. This is the result:

How many disks?
3
move disk 1 from post 1 to post 3
move disk 2 from post 1 to post 2

move disk 1 from post 3 to post 2
move disk 3 from post 1 to post 3
move disk 1 from post 2 to post 1
move disk 2 from post 2 to post 3
move disk 1 from post 1 to post 3

And here is the code of  the main program, and the subroutine tower with the usability changes 
highlighted in red.

!      mayer & perkins
!      "towers of hanoi revisited"

       program hanoi
       integer :: disks
       do
    10     format('How many disks? ')
           write(*,10)
           read (*,"(i10)") disks
           if (disks > 0) then
               call tower(disks)
               exit
           else if (disks == 0) then
               exit
           end if
       end do
       end

       subroutine tower(total)
       integer, intent(in) :: total
       integer :: from, tod, disk, temp, i
       integer :: bitstr(10), hold(10)
       character(len=6), dimension(3) :: place
       place = (/'post 1', 'post 2', 'post 3'/)
       do i = 1,10
           bitstr(i) = 0
           hold(i) = 1
       end do
       temp = 3 - mod(total,2)
       do i = 1, 2**total - 1
           disk = incr(bitstr)
           if (disk == 1) then
               from = hold(1)
               tod = 6 - from - temp
               temp = from
           else
               from = hold(disk)
               tod = 6 - hold(1) - hold(disk)
           end if
           write(*,20) 'move disk ',disk,' from ',place(from),' to ',place(tod)
           hold(disk) = tod
       end do
    20 format(A,I1,A,A,A,A)
       end

Note that in the formatting string A defines an ASCII string, and I an integer. Just putting I 
instead of  I1 will lead to a compiler error.

As a final note, the original indentation associated with the fixed formatting has not been 
removed here - as the indentation of  the whole program conforms to the standard of  4 spaces, 
changing the left-justification is personal preference. The program is small enough to leave it as 
is.

See listing hanoi_6.f95.

